An Overview of ODS Statistical Graphics in SAS® 9.4

Robert N. Rodriguez and Warren F. Kuhfeld, SAS Institute Inc., Cary, NC

ABSTRACT

ODS Statistical Graphics (ODS Graphics for short) provides functionality for creating statistical graphics. ODS Graphics is available in the Base SAS®, SAS/STAT®, SAS/ETS®, and SAS/QC® products. More than 100 statistical procedures use this functionality, and they produce graphs as automatically as they produce tables. In addition, Base SAS procedures use ODS Graphics to produce plots for exploratory data analysis and for customized statistical displays.

This paper presents the essential information that you need to get started with ODS Graphics in SAS® 9.4. ODS Graphics is an extension of ODS (the Output Delivery System), which manages procedure output for display in a variety of destinations, such as HTML and RTF. Consequently, many familiar features of ODS for tabular output apply equally to graphs.

When ODS Graphics is enabled, graphs and tables are integrated together in your ODS output destination. ODS Graphics produces graphs in standard image file formats, and the consistent appearance and individual layout of these graphs are controlled by ODS styles and templates, respectively.

Because the default templates for procedure graphs are provided by SAS software, you do not need to know the details of templates in order to create statistical graphics. However, with some understanding of the underlying Graph Template Language, you can modify the default templates to make changes to graphs that are permanently in effect each time you run the procedure. Alternatively, you can make immediate changes by using the ODS Graphics Editor, whose point-and-click interface enables you to customize titles, annotate points, and make other enhancements.

INTRODUCTION

Effective graphs are indispensable for modern statistical analysis. They reveal patterns, differences, and uncertainty that are not readily apparent in tabular output. Graphs provoke questions that stimulate deeper investigation, and they add visual clarity and rich content to reports and presentations.

The only SAS licenses that are required in order to use ODS Graphics are a Base SAS license and a license for the product that contains the procedure that you are using. (See the list on page 12.)

Depending on a number of factors, ODS Graphics might or might not be enabled by default. ODS Graphics is usually enabled by default in the SAS windowing environment; it is usually disabled by default when you invoke SAS in other ways. However, these defaults can be changed in a number of ways. You can change the default in the SAS windowing environment by selecting **Tools ► Options ► Preferences** from the menu at the top of the main SAS window. Then on the **Results** tab, select the **Use ODS Graphics** check box to enable ODS Graphics by default, or clear the check box to disable ODS Graphics by default.

You can also enable ODS Graphics by specifying the following statement:

```
ods graphics on;
```

Usually, ODS Graphics is enabled once at the beginning of your SAS session, and it stays enabled for the duration of the session. You might want to consider disabling ODS Graphics if your goal is solely to produce computational or tabular results, because ODS Graphics uses additional resources. Otherwise, you do not need to disable and then re-enable ODS Graphics between procedure steps.

When ODS Graphics is enabled, statistical procedures that support ODS Graphics create appropriate graphs, either by default or when you specify procedure options for requesting specific graphs. These options are documented in the "Syntax" section of the procedure chapters in the SAS/STAT, SAS/ETS, and SAS/QC user's guides and in the Base SAS statistical procedures guide. The "Details" section of the procedure chapters includes a subsection titled "ODS Graphics" that lists the available graphs, and many of these graphs are illustrated in the "Examples" section.

The following statements illustrate how you can create a default plot for a simple linear regression analysis:

```
proc reg data=sashelp.Class;
   model Weight=Height;
quit;
```

Figure 1 shows the output produced by the REG procedure for this example. The output consists of a table of parameter estimates and a fit plot. Both the table and the plot are part of the default output (not shown here) produced by the REG procedure for a simple linear regression analysis. The default graphs also include regression diagnostics plots and a residuals plot. Note that the fit plot is accompanied by an inset that provides information relevant to the fit.

Figure 1 SAS Output Using the HTMLBLUE Style

The REG Procedure
Model: MODEL1
Dependent Variable: Weight

Parameter Estimates

Variable	DF	Parameter Estimate	Standard Error	t Value	Pr > |t|
Intercept	1	-143.02692	32.27459	-4.43	0.0004
Height	1	3.89903	0.51609	7.55	<.0001



This example demonstrates how procedures that support ODS Graphics take advantage of computational results to enrich their graphs. With traditional graphics, creating a fit plot such as this one would require hundreds of lines of additional SAS program statements.

The output in Figure 1 is displayed in the default ODS style for the HTML destination, which is HTMLBLUE. ODS styles control the colors, fonts, and general appearance of all graphs and tables.

SAS provides other styles that are recommended for use with statistical graphics. The JOURNAL style is a gray-scale style that is especially useful for graphs that will appear in journals and other black-and-white publications. The following statements use the JOURNAL style to produce PDF output for the regression example:

```
ods pdf style=journal;
proc reg data=sashelp.Class;
   model Weight=Height;
quit;
ods pdf close;
```

Figure 2 shows the output.

Figure 2 SAS Output Using the JOURNAL Style

The REG Procedure
Model: MODEL1
Dependent Variable: Weight

Parameter Estimates

Variable	*DF*	*Parameter Estimate*	*Standard Error*	*t Value*	*Pr > |t|*
Intercept	1	-143.02692	32.27459	-4.43	0.0004
Height	1	3.89903	0.51609	7.55	<.0001



GRAPH GALLERY

Statistical procedures that use ODS Graphics produce a rich variety of graphs. This section illustrates a few of the many graphs that are available, either by default or when requested by the PLOTS= option, which is usually available in the procedure statement. For more information, see the “Syntax” and “ODS Graphics” sections of the procedure chapters in the user’s guides. The graphs that are shown here are created using the HTMLBLUE style for visual consistency.

Figure 3 displays a default panel of regression diagnostic plots that is created by the REG procedure. Similar diagnostic panels are available from other linear modeling procedures, and the plots in the panels can be displayed individually.

Figure 4(a) displays a default plot of response values versus covariate values that the GLM procedure creates when you specify an analysis of covariance model that has one or two CLASS variables and one continuous variable. The

lines represent the fitted relationship within each classification level. Figure 4(b) displays a default grouped box plot of response values that the GLM procedure creates when you specify a one-way analysis of variance model.

Figure 4(c) and Figure 4(d) display optional surface and contour plots for nonparametric bivariate density estimates that are created by the KDE procedure.

Figure 4(e) displays an optional prediction plot that is created by the KRIGE2D procedure. Figure 4(f) displays an enhanced Shewhart chart that is created by the SHEWHART procedure.

Figure 5 displays an optional scatter plot matrix of scores that is created by the PRINCOMP procedure.

The next section explains the basics of using ODS Graphics to create and manage graphs. For additional resources, see the section "RECOMMENDED READING" on page 26.

Figure 3 REG Procedure



Figure 4 Graph Gallery

(a) GLM Procedure



(b) GLM Procedure



(c) KDE Procedure



(d) KDE Procedure



(e) KRIGE2D Procedure



(f) SHEWHART Procedure



Figure 5 PRINCOMP Procedure



A PRIMER ON ODS STATISTICAL GRAPHICS

You can enable ODS Graphics by specifying the following statement:

```
ods graphics on;
```

ODS Graphics remains enabled for all procedure steps until you disable it by specifying the following statement:

```
ods graphics off;
```

As explained later in this paper, you use ODS GRAPHICS statement options to specify characteristics of your graphs, such as size, image format, and image filename. For more information, see the section "Syntax" in Chapter 21, "Statistical Graphics Using ODS," in the *SAS/STAT User's Guide.*

After you have enabled ODS Graphics, creating graphical output from procedures is as simple as creating tabular output. You can control your output in the following ways:

- ODS destination statements (such as ODS HTML or ODS RTF) specify where your graphs are displayed. For a list of the supported destinations, see the section "ODS Destination Statements" in Chapter 21, "Statistical Graphics Using ODS," in the *SAS/STAT User's Guide*.
- Procedure options and defaults determine which graphs are created. For procedures that support ODS Graphics, these options are described in the "Syntax" section of the corresponding procedure chapter in the SAS/STAT, SAS/ETS, and SAS/QC user's guides and in the Base SAS statistical procedures guide. You usually request nondefault graphs by specifying the PLOTS= option in the procedure statement.
- ODS SELECT and ODS EXCLUDE statements select and exclude graphs from your output. As with tables, you refer to graphs by name in these statements. In the procedure chapters, the names of available graphs are listed in the "ODS Graphics" subsection of the "Details" section.

- The ODS OUTPUT statement creates SAS data sets from the data objects that are used to make plots.
- ODS styles control the general appearance and consistency of all graphs and tables.
- ODS templates modify the layout and details of each graph. For more information, see Chapter 21, "Statistical Graphics Using ODS," and Chapter 22, "ODS Graphics Template Modification," in the *SAS/STAT User's Guide.*

  NOTE: A default template is provided for each graph, so you do not need to know anything about templates in order to create statistical graphics.

You can also access individual graphs, control the resolution and size of graphs, and modify your graphs as explained later in this paper.

Graph Styles

ODS styles control the overall appearance of graphs and tables. They specify colors, fonts, line styles, symbol markers, and other attributes of graph elements. There are two types of ODS styles: one type distinguishes groups of observations by colors, markers, and lines; the other type distinguishes groups of observations by colors only. This distinction is called attribute priority:

- An ATTRPRIORITY=NONE style distinguishes groups of observations by colors, markers, and lines. (See Figure 10.) Most ODS styles are ATTRPRIORITY=NONE styles. They are compromise styles in the sense that some graph elements are intentionally overdistinguished to facilitate black-and-white printing. For example, fit lines that correspond to different classification levels are distinguished by both colors and line patterns.
- An ATTRPRIORITY=COLOR style distinguishes groups of observations by colors and not by lines or symbols.[1] (See Figure 4(a).) ATTRPRIORITY=COLOR styles include HTMLBLUE, PEARL, PEARLJ, and SAPPHIRE.

If you want to control the markers or lines that are displayed for groups of observations when you are using an ATTRPRIORITY=COLOR style, first be sure to specify the ATTRPRIORITY=NONE option in the ODS GRAPHICS statement. For more information, see the sections "Attribute Priorities" and "Overriding How Groups Are Distinguished" in Chapter 21, "Statistical Graphics Using ODS," in the *SAS/STAT User's Guide.*

Although you can use any ODS style, only a few styles are usually used with ODS Graphics. They are described in Table 1.

You specify an ODS style by using the STYLE= option in the ODS destination statement. For example, the following statement creates RTF output and specifies the JOURNAL style:

```
ods rtf style=Journal;
```

The following statement sets the style for the LISTING destination:

```
ods listing style=HTMLBlue;
```

The style that is specified by the STYLE= option in the ODS LISTING statement applies only to graphs. SAS monospace format is used for tables.

More generally, you can modify the colors, fonts, and other attributes of graph elements in an ODS style by editing the style template. For more information, see the section "ODS Styles" in Chapter 21, "Statistical Graphics Using ODS," in the *SAS/STAT User's Guide,* and see the *SAS Output Delivery System: User's Guide*.

[1] More precisely, an ATTRPRIORITY=COLOR style such as HTMLBLUE distinguishes the first 12 groups of observations only by color. Markers and lines change for groups 13–24 and then again for groups 25–36.

Table 1 ODS Styles Most Often Used with ODS Graphics

Style	Recommended Destinations	Attribute Priority	Description
ANALYSIS	HTML	None	Color style, with sans serif fonts, whose dominant colors are yellow, green, and tan
DEFAULT	HTML	None	Color style, with bold sans serif fonts, whose dominant colors are gray, blue, and white
HTMLBLUE ✓	HTML	Color	Color style, with sans serif fonts, whose dominant colors are shades of blue. Default for HTML destination and SAS/STAT documentation
HTMLBLUECML	HTML	None	ATTRPRIORITY="None" version of HTMLBLUE
JOURNAL, ✓ JOURNAL1A✓	PDF, PS, RTF, PRINTER	None	Black-and-white styles with sans serif fonts and filled areas
JOURNAL2, ✓ JOURNAL2A ✓	PDF, PS, RTF, PRINTER	None	Black-and-white styles, similar to JOURNAL but with empty areas. Grouped bar charts use crosshatching to show groups.
JOURNAL3, ✓ JOURNAL3A ✓	PDF, PS, RTF, PRINTER	None	Black-and-white styles, similar to JOURNAL2 but with a mix of filled areas and crosshatching in grouped bar charts
LISTING	HTML, LISTING	None	Color style, similar to DEFAULT but with a white background. Default for the LISTING destination.
PEARL ✓	PDF, PS, RTF, PRINTER	Color	Color style, with sans serif fonts and a white background, whose dominant colors are shades of blue. Default for PDF destination.
PEARLJ ✓	PDF, PS, RTF, PRINTER	Color	Color style, with sans serif fonts and a white background, whose dominant colors are shades of blue. Default for PDF tables in SAS/STAT documentation.
RTF	RTF	None	Color style, with serif (Times Roman) fonts, whose dominant colors are blue, white, and black. Default for RTF destination.
SAPPHIRE ✓	PDF, PS, RTF, PRINTER	Color	Color style, with sans serif fonts, a white background, and a light blue table heading background, whose dominant colors are shades of blue
STATISTICAL	HTML	None	Color style, with sans serif fonts, whose dominant colors are blue, gray, and white

✓Newer styles that are recommended for use with statistical graphics.
JOURNAL# styles differ from JOURNAL#A styles in that the former use italic in table headings.

Accessing Individual Graphs

For most ODS destinations (including HTML, RTF, and PDF), graphs and tables are integrated together in the output, and you view your output by using an appropriate viewer, such as a web browser for HTML.

A graph is created for every open destination. When you open a new destination, you should close all destinations that you do not need. This makes your jobs run faster and consume fewer resources, because fewer graphs are produced.

If you are writing a paper or creating a presentation, you need to access your graphs individually. There are various ways to do this, depending on the ODS destination. Three methods are particularly useful:

- If you are viewing RTF output, you can copy and paste your graphs from the viewer into a Microsoft Word document or a Microsoft PowerPoint slide.
- If you are viewing HTML output, you can copy and paste your graphs from the viewer, or you can right-click on the graph and save it to a file. Note that RTF has a higher default resolution than HTML. You can increase the

resolution in HTML by specifying the ODS HTML IMAGE_DPI=300 statement. For more information, see the section "Specifying the Size and Resolution of Graphs" on page 9.

- You can save your graphs in image files and then include them in a paper or presentation. For example, you can save your graphs as PNG (portable network graphics) files and include them in a paper that you are writing in LaTeX or HTML.

You can specify the graphics image format and the filename in the ODS GRAPHICS statement. For example, the following statements, when submitted before a procedure step that produces multiple graphs, save the graphs in PostScript files named *myname.ps*, *myname1.ps*, and so on:

```
ods listing close;
ods latex file="test.tex" path="C:\myfiles" gpath="C:\myfiles\ps";
ods graphics on / imagefmt=ps imagename="myname";
```

The section "Image File Types" in Chapter 21, "Statistical Graphics Using ODS," in the *SAS/STAT User's Guide* provides details about the file types available for various destinations, how they are named, and how they are saved.

If you are using the LISTING destination and the SAS windowing environment, you can also copy your graphs from the default image viewer and paste them into a Microsoft Word document or a Microsoft PowerPoint slide.

Specifying the Size and Resolution of Graphs

Two factors to consider when you are creating graphs for a paper or presentation are the size of your graphs and their resolution. For best results, you should specify the size of the graph as it will appear in the document (rather than resizing the graph after it has been produced). You can specify the size in the ODS GRAPHICS statement, as illustrated by the following examples:

```
ods graphics on / width=6in;
ods graphics on / height=4in;
ods graphics on / width=4.5in height=3.5in;
```

When only one dimension is specified, most graphs are produced by using a default width/height aspect ratio of 4/3.

The default resolution of graphs that are created for the HTML and LISTING destinations is 100 DPI (dots per inch), whereas the default for the RTF destination is 200 DPI. You can change the resolution by specifying the IMAGE_DPI= option in any ODS destination statement, as in the following example:

```
ods html image_dpi=300
```

An increase in resolution often improves the quality of the graphs, but it also increases the size of the image file. The section "Graph Size and Resolution" in Chapter 21, "Statistical Graphics Using ODS," in the *SAS/STAT User's Guide* provides more information about graph size and resolution.

MODIFYING YOUR GRAPHS

Although ODS Graphics is designed to automate the creation of high-quality statistical graphics, you might occasionally need to modify your graphs. There are two ways to proceed:

- Use the ODS Graphics Editor, which provides a point-and-click interface, to make changes that are data-dependent and immediate. This approach is recommended if you are making ad hoc changes to a *specific* graph that you have created and are preparing for a paper or presentation.
- Edit the ODS graph template for a plot to make changes that are persistent—in other words, applied *each time you run the procedure*.

The next two sections discuss these approaches.

Modifying Your Graphs by Using the ODS Graphics Editor

You can use the ODS Graphics Editor to customize titles and labels, annotate data points, add text, and change graph element properties such as fonts, colors, and line styles. After you have modified your graph, you can save it as a PNG image file or as an SGE file, a special SAS file type that preserves the editing context. You can open previously saved SGE files by using the ODS Graphics Editor and resume editing.

1. Temporarily Enable Creation of Editable Graphs by Using an ODS Destination Statement

You can enable the creation of editable graphs within a SAS session by submitting one of the following statements:

```
ods listing sge=on;
ods html sge=on;
```

You can disable the creation of editable graphs by submitting one of the following statements:

```
ods listing sge=off;
ods html sge=off;
```

You can use other destinations instead.

2. Permanently Enable Creation of Editable Graphs across SAS Sessions

You can create a default setting that enables or disables the creation of editable graphs across SAS sessions via the "ODS Graphics Editor" setting in the SAS Registry. You can change this setting in the SAS windowing environment as follows:

1 Open the Registry Editor by entering **regedit** on the command line.
2 Select **SAS_REGISTRY ► ODS ► GUI ► RESULTS**.
3 In the **Name** field, click **ODS Graphics Editor** to open the Edit String Value dialog box.
4 Type **On** to enable the creation of editable graphs or **Off** to disable it, and click **OK**.

Creating editable graphs uses additional resources, so you might not want to permanently enable this feature.

Figure 6 shows how to invoke the ODS Graphics Editor. The second graph in each identically labeled graph pair is the editable graph. Figure 7 shows the ODS Graphics Editor window for the editable diagnostic plot created by the ROBUSTREG procedure. In Figure 8, various tools in the ODS Graphics Editor have been used to modify the title and annotate a particular point. You can save the edited plot as a PNG file or as a re-editable SGE file by selecting **File ► Save As**.

Figure 6 Invoking the ODS Graphics Editor for an Editable Plot



Figure 7 Diagnostic Plot before Editing



Figure 8 Diagnostic Plot after Editing



For more information about the tools available in the ODS Graphics Editor, see the *SAS 9.4 ODS Graphics Editor: User's Guide*. Note that the ODS Graphics Editor does not permit you to make structural changes to a graph (such as moving the positions of data points). It provides you with a point-and-click interface to make one-time changes to a specific graph, whereas modifying the graph template, discussed in the next section, provides you with a programmatic way to make template changes that persist every time you run the procedure.

Modifying Your Graphs by Editing Graph Templates

Graphs that are produced by ODS Graphics are constructed from two underlying components: a data object supplied by a procedure at run time and a compiled graph template that is designed to work with this data object. Together, the data object and the template form an output object that ODS displays in one or more output destinations. Each output destination uses an ODS style.

A graph template is a program, written in the Graph Template Language (GTL), that specifies the layout and details of a graph. SAS provides a template for every graph that is produced by all statistical procedures. These templates are complete descriptions of how the graphs are to be produced. Consequently, most of the default graph templates that SAS provides are lengthy and complex.

Ordinarily, you do not need to know anything about templates or the GTL in order to create graphs by running procedures that use ODS Graphics. However, if you have a moderate understanding of the GTL, you can edit graph templates to make changes that are applied when you rerun the procedure. For instance, you can modify titles and axis labels, and you can add footnotes that contain project information. Example 2 on page 17 illustrates this type of modification. For more information, see Chapter 22, "ODS Graphics Template Modification," in the *SAS/STAT User's Guide.* A second reason for learning the GTL is that you can use it to create highly customized displays by writing your own graph templates; for more information, see the section "GENERAL-PURPOSE STATISTICAL GRAPHICS PROCEDURES" on page 13.

STATISTICAL PROCEDURES THAT SUPPORT ODS GRAPHICS IN SAS 9.4

The following statistical procedures support ODS Graphics in SAS 9.4:

SAS/STAT

ACECLUS
ADAPTIVEREG
ANOVA
BCHOICE
BOXPLOT
CALIS
CLUSTER
CORRESP
FACTOR
FMM
FREQ
GAM
GAMPL
GEE
GENMOD
GLIMMIX
GLM
GLMPOWER
GLMSELECT
HPFMM
HPSPLIT
ICLIFETEST
ICPHREG
IRT
KDE
KRIGE2D
LIFEREG
LIFETEST
LOESS
LOGISTIC
MCMC
MDS
MI
MIXED
MULTTEST
NLIN
NPAR1WAY
ORTHOREG
PHREG
PLM
PLS
POWER
PRINCOMP
PRINQUAL
PROBIT
QUANTLIFE
QUANTREG
QUANTSELECT
REG
ROBUSTREG
RSREG
SEQDESIGN
SEQTEST
SIM2D
SPP
STDRATE
SURVEYFREQ
SURVEYLOGISTIC
SURVEYMEANS
SURVEYPHREG
SURVEYREG
TPSPLINE
TRANSREG
TTEST
VARCLUS
VARIOGRAM

Base SAS

CORR
FREQ
UNIVARIATE

SAS/QC

ANOM
CAPABILITY
CUSUM
MACONTROL
MVPDIAGNOSE
MVPMONITOR
MVPMODEL
PARETO
RAREEVENTS
RELIABILITY
SHEWHART

Other

HPF
HPFENGINE

SAS Risk Dimensions

SAS/ETS

ARIMA
AUTOREG
COPULA
COUNTREG
ENTROPY
ESM
EXPAND
HPCDM
HPQLIM
HPSEVERITY
MODEL
PANEL
PDLREG
QLIM
SEVERITY
SIMILARITY
SSM
SYSLIN
TIMEDATA
TIMEID
TIMESERIES
UCM
VARMAX
X12

For more information about the specific graphs available from a particular procedure, see the "Syntax" and "ODS Graphics" sections of the corresponding procedure chapter in the SAS/STAT, SAS/ETS, and SAS/QC user's guides and the Base SAS statistical procedures guide.

PROCEDURES THAT SUPPORT ODS GRAPHICS AND TRADITIONAL GRAPHICS

A number of procedures that support ODS Graphics also produced traditional graphics in previous SAS releases. These include the UNIVARIATE procedure in Base SAS; the BOXPLOT, LIFEREG, LIFETEST, and REG procedures in SAS/STAT; and the ANOM, CAPABILITY, CUSUM, MACONTROL, PARETO, RELIABILITY, and SHEWHART procedures in SAS/QC. All these procedures continue to produce traditional graphics, but in some cases they do so only when ODS Graphics is not enabled. For more information about the interaction between traditional graphics and ODS graphics in these procedures, see the chapters for these procedures in their respective user's guides.

Note that traditional graphs are saved in SAS graphics catalogs and are controlled by the GOPTIONS statement. In contrast, ODS Graphics produces graphs in standard image file formats (not graphics catalogs), and the appearance and layout of these graphs are controlled by ODS styles and templates, respectively.

GENERAL-PURPOSE STATISTICAL GRAPHICS PROCEDURES

Statistical procedures that support ODS Graphics create graphs in the context of a specific analysis. However, statistical graphics are also essential for exploring data and for constructing specialized displays for novel analyses. These situations require general-purpose graphical tools for creating stand-alone plots.

Base SAS provides a family of statistical graphics procedures that are designed to meet these needs. The following procedures use ODS Graphics functionality and provide a convenient syntax for creating a variety of plots directly from data:

- PROC SGPLOT creates single-cell plots that contain a variety of plot and chart types. See Figure 14 for an example.
- PROC SGSCATTER creates single-cell and multicell scatter plots and scatter plot matrices that contain optional fits and ellipses. See Figure 17 for an example.
- PROC SGPANEL creates single-page or multipage panels of plots and charts conditional on classification variables. See Figure 11 for an example.

These procedures, which are collectively referred to as the "SG procedures," can produce many types of plots, including density plots, dot plots, needle plots, series plots, horizontal and vertical bar charts, histograms, and box plots. They can also compute and display loess fits, polynomial fits, penalized B-spline fits, reference lines, bands, and ellipses. Graphs that are produced by the SG procedures and statistical procedures have a consistent appearance that is determined by the ODS style. The SG procedures are documented in the *SAS ODS Graphics: Procedures Guide*.

For situations that require highly customized displays not available from the SG procedures, you can write your own graph templates, taking advantage of the power of the Graph Template Language. You can then apply these templates to your data and render the graphs by using the SGRENDER procedure. This use of the GTL is outside the scope of this paper, but Chapter 22, "ODS Graphics Template Modification," in the *SAS/STAT User's Guide* provides an overview and examples. For complete documentation of the Graph Template Language, see the *SAS Graph Template Language: Reference* and the *SAS Graph Template Language: User's Guide*.

You do not need to enable ODS Graphics in order to use the SG procedures. However, the options available in the ODS GRAPHICS statement are applicable to these procedures.

See Figure 9 for examples of clinical graphs that are created by using the SG procedures. For more information, see Matange (2016).

ODS GRAPHICS DESIGNER

You can use the ODS Graphics Designer, which provides a point-and-click interface, to create graphs without typing any syntax. You can use the ODS Graphics Designer in many ways. For example, you can start by selecting **Tools ► ODS Graphics Designer** and then **File ► New ► Blank Graph**. You can drag a graph type such as **Scatter** into the blank graph. Select a data set such as **Sashelp** and **Baseball**. Designate X- and Y-axis variables such as **CrRBI** (career runs batted in) and **LogSalary** (log salary). Designate **Div** as a group variable.[2] Then click **OK**. You can drag a **Discrete Legend** into the top right corner. You can finish by dragging a **Cell Header** to the top of the graph and typing the title. The results are displayed in Figure 10.

[2] **Div** values begin with an "A" or "N" for American or National League, followed by an "E" or "W" for East or West.

Figure 9 Gallery of Clinical Graphs Created by SG Procedures



















Figure 10 ODS Graphics Designer



EXAMPLE 1: STATISTICAL GRAPHICS FOR A LINEAR MODEL ANALYSIS

This example illustrates the start-to-finish use of ODS Graphics in an analysis in which the SGPANEL procedure creates a preliminary display of the data and the SAS/STAT GLM procedure creates a specialized display that adds information to the statistical analysis.

The following statements create a data set that contains a response variable **y** and two classification variables, **a** and **b**:

```
data measure;
   drop i abEffect;
   do a = 1 to 3;
      do b = 1 to 3;
         if ((a = 3) & (b = 3)) then abEffect = 3;
         else                        abEffect = 1;
         do i = 1 to 10;
            y = abEffect + rannor(1);
            output;
         end;
      end;
   end;
run;

proc sort data=measure; by b;
run;
```

The next statements use the SGPANEL procedure to plot the means of **y** for levels of **a** in a display that is paneled by the levels of **b**:

```
title "Two-Factor Interaction Plot";
proc sgpanel data=measure;
   panelby b / columns=3 spacing=5;
   vline a / response=y stat=mean limits=both markers
             legendlabel= "Cell Means with 95% Confidence Limits";
   discretelegend;
run;
title;
```

This is often referred to as a “means plot” or a “two-factor interaction plot.” The display, shown in Figure 11, suggests the presence of an interaction effect, which should be included in a follow-up analysis of variance (ANOVA).

Figure 11 Interaction Plot Produced by the SGPANEL Procedure



The next statements perform the analysis of variance by using the GLM procedure:

```
proc glm data=measure;
   ods select ModelANOVA DiffPlot;
   class a b;
   model y = a|b / ss3;
   lsmeans a*b / pdiff=all;
quit;
```

The LSMEANS statement requests least squares means (LS-means) for the interaction of **a** and **b**. The option PDIFF=ALL requests p-values for all pairwise differences of the LS-means.The ANOVA table, shown in Figure 12, indicates that the main and interaction effects are significant.

Figure 12 ANOVA Results from the GLM Procedure

The GLM Procedure

Dependent Variable: y

Source	DF	Type III SS	Mean Square	F Value	Pr > F
a	2	11.19595624	5.59797812	6.42	0.0026
b	2	18.09492619	9.04746309	10.37	<.0001
a*b	4	14.27143232	3.56785808	4.09	0.0045

When you specify an LSMEANS statement and the PDIFF= option, PROC GLM produces a default plot appropriate for the LS-means comparison. For PDIFF=ALL, the procedure produces a diffogram, which displays all pairwise LS-means differences and indicates which ones are significant. The diffogram, shown in Figure 13, gives you more information about the interaction by telling you which of the cell mean differences are significant.

Figure 13 LS-Means Diffogram Produced by the GLM Procedure



This display is related to the mean-mean scatter plot of Hsu and Peruggia (1994); also see Hsu (1996). Diffograms are also available from the GLIMMIX procedure. For more information and examples, see Chapter 46, "The GLIMMIX Procedure," in the *SAS/STAT User's Guide.*

EXAMPLE 2: STATISTICAL GRAPHICS FOR A SURVIVAL STUDY

This example illustrates ODS Graphics by using a combination of SG procedures and statistical procedures in the context of a survival study. The **UIS** data set that is used here is a subset of data from the University of Massachusetts Amherst Aids Research Unit Impact Study (UIS). This study, which consisted of two concurrent randomized trials of residential treatment for drug abuse, compared treatment programs of different planned durations designed to reduce drug abuse. For additional background, see Hosmer, Lemeshow, and May (2008).

The following statements create the **UIS** data set, which contains 628 observations and 12 variables:

```
data UIS;
   input ID Age BeckTotal HerCoc IVHx nDrugTx Race Treat Site Lot Time Censor;
   label censor = "Returned to Drug Use"
         time   = "Time to Return to Drug Use (Days)"
         treat  = "Treatment"
         site   = "Treatment Site";
   if ivhx=2 then ivhx=1;
   nDrugFP1 = 1/((ndrugtx+1)/10);
   nDrugFP2 = 1/((ndrugtx+1)/10) * log((ndrugtx+1)/10);
   datalines;
1   39  9 4 3 1 0 1 0 123 188 1
2   33 34 4 2 8 0 1 0  25  26 1

   ... more lines ...

628 46 31.5 1 3 15 1 1 1   9 377 1
;
```

The response variable **Time** is the survival time in days (measured from admission) that a participant avoids returning to drug use. The censoring indicator variable **Censor** has a value of 1 if the participant returned to drug use, and a value of 0 otherwise. For each participant, the variable **Treat** indicates the treatment randomization assignment (0=Short and 1=Long), and the variable **Site** indicates the treatment site (0=A and 1=B). There were 444 participants at Site A who received either a 3-month (short) or 6-month (long) treatment. There were 184 participants at Site B who received either a 6-month (short) or 12-month (long) treatment.

The following statements group the data so that you can examine the times by combinations of treatment and site:

```
data grouped;
   set UIS;
   length EventType $ 9;
   label  EventType="Type";
   if      (treat=0 and site=0) then do; Group="Short*Site A"; Treatment="Short"; end;
   else if (treat=0 and site=1) then do; Group="Short*Site B"; Treatment="Short"; end;
   else if (treat=1 and site=0) then do; Group="Long*Site A";  Treatment="Long";  end;
   else do;                              Group="Long*Site B";  Treatment="Long";  end;
   if (censor=0) then EventType = 'Censored';
   else               EventType = 'Event';
run;
```

The variable **EventType** indicates whether a participant returned to drug use.

Plot the Survival Times for All Treatment × Site Combinations

The next statements use the SGPLOT procedure to plot the survival times for all combinations of treatment and site:

```
proc sgplot data=grouped;
   title "Survivorship Data for Long and Short Treatments";
   footnote justify=left "Source: Hosmer, Lemeshow, and May,"
            italic " Applied Survival Analysis";
   xaxis grid label = "Time to Return to Drug Use (Days)";
   yaxis      label = " ";
   scatter x=time y=Group / group=EventType transparency=0.5;
run;

title;
footnote;
```

The SGPLOT procedure plots **Time** for the four combinations of treatment and site, as shown in Figure 14. Note that the ODS GRAPHICS ON statement is not required when you use the SG procedures. Values of **Group** are displayed along the Y axis, and values of **Time** are displayed along the X axis. The GROUP= option differentiates groups of observations by using a different color for each level of **EventType**. The actual colors (blue for Event and red for Censored) are determined by the HTMLBLUE style. The TRANSPARENCY= option controls the density of the circles by specifying their degree of transparency. Figure 14 shows that the longer survival times are mostly censored.

Figure 14 Display of Survival Times Produced by the SGPLOT Procedure



Use PROC LIFETEST to Plot Survival Curves

Specialized methods are needed to analyze the data. The product-limit (Kaplan-Meier) estimator is commonly used to describe the survivor function. The following statements use the LIFETEST procedure to create a plot of two product-limit functions for each site (one for each treatment program). First, the data are sorted by the variable **Site** so that PROC LIFETEST can create survival curves for different sites in separate plots:

```
proc sort data=UIS;
   by site;
run;

proc lifetest data=grouped plots=survival(cb=hw test atrisk=0 to 1500 by 250);
   ods select SurvivalPlot;
   time time * censor(0);
   strata Treatment;
   by site;
run;
```

The PLOTS= option requests a graph of product-limit curves, 95% Hall-Wellner confidence bands, and the number of subjects at risk. At-risk times are specified in the ATRISK= option.

Figure 15 shows the graph of the estimated survivor functions for Site A, which reveals that participants in the long program survive longer than participants in the short program. In other words, participants in the short program return to drug use earlier than those in the long program. The numbers of subjects at risk shown in Figure 15 are consistent with this conclusion. The p-value for the log-rank test (p=0.0047) indicates a significant difference in the survivorship between participants in the long and short programs. Similar conclusions can be drawn from the display for Site B (not shown here).

Figure 15 Survival Plot for Site A Produced by the LIFETEST Procedure



Customize the PROC LIFETEST Graph Template

The LIFETEST procedure, like other statistical procedures that use ODS Graphics, provides a PLOTS= option for modifying its graphs. When these options are not sufficient, you can modify the template for a graph to make changes that persist each time you run the procedure. For more information, see Chapter 22, "ODS Graphics Template Modification," and Chapter 23, "Customizing the Kaplan-Meier Survival Plot," in the *SAS/STAT User's Guide*. You begin by downloading from the web some macros that are specifically designed to make it easy to modify the Kaplan-Meier plot:

```
data _null_;
   %let url = //support.sas.com/documentation/onlinedoc/stat/ex_code/141;
   infile "http:&url/templft.html" device=url;

   file 'macros.tmp';
   retain pre 0;
   input;
   if index(_infile_, '</pre>') then pre = 0;
   if pre then put _infile_;
   if index(_infile_, '<pre>')  then pre = 1;
run;

%inc 'macros.tmp' / nosource;
```

The next statements modify the template by changing the default title of the survival plot from "Product-Limit Survival Estimate" to "Kaplan-Meier Plot":

```
%ProvideSurvivalMacros
%let TitleText2 = "Kaplan-Meier Plot";
%CompileSurvivalTemplates

ods graphics on / byline=footnote;
proc lifetest data=grouped plots=survival(cb=hw test atrisk=0 to 1500 by 250);
   ods select SurvivalPlot;
   time time * censor(0);
   strata Treatment;
   by site;
run;
```

This example uses a modularized template that is available in the SAS sample library. Complete details are provided in Chapter 23, "Customizing the Kaplan-Meier Survival Plot," in the *SAS/STAT User's Guide.* The modified survival plots are shown in Figure 16. This particular modification can also be made by the ODS Graphics Editor, which is recommended for making one-time changes to a graph as opposed to making persistent changes.

The following statements delete the modified template from SASUSER.TEMPLAT and revert to the default template in SASHELP.TMPLMST, which is where the SAS templates are stored:

```
proc template;
   delete Stat.Lifetest.Graphics.ProductLimitSurvival / store=sasuser.templat;
run;
```

The section "Controlling Output Appearance with Templates" in Chapter 20, "Using the Output Delivery System," in the *SAS/STAT User's Guide* describes how you can use the ODS PATH statement along with the template libraries that you create.

Use PROC SGSCATTER to Display Diagnostics Plots

In order to account for the risk factors in the UIS data, the following statements fit a proportional hazards model and create an output data set that contains 10 weighted Schoenfeld residual variables, **W1**–**W10**, one for each effect:

```
proc phreg data=grouped noprint;
   class ivhx(ref="1");
   model time * censor(0) = age becktotal ndrugfp1 ndrugfp2 ivhx race treat
                            site age*site race*site;
   output out=out1 wtressch=w1-w10;
run;
```

The output from the PHREG procedure is not displayed. You can evaluate the adequacy of the model by plotting the weighted Schoenfeld residuals against the survival times for each independent variable. Because the survival times are skewed, it is helpful to apply a log transformation to **Time** before making the plots. This is done using the following statements:

```
data plot;
   set out1;
   LogTime = log(time);
   label logtime = "log(Time)";
run;
```

Figure 16 Modified Survival Plots Produced by the LIFETEST Procedure





The following statements use the SGSCATTER procedure to create the scatter plots:

```
proc sgscatter data=plot;
   title 'Weighted Schoenfeld Residuals and Loess Smooths';
   label w1  = "Age"          w2  = "BeckTotal"
         w3  = "nDrugFP1"     w4  = "nDrugFP2"
         w5  = "IVHx3"        w6  = "Race"
         w7  = "Treatment"    w8  = "Site"
         w9  = "AgeSite"      w10 = "Race * Site";

   plot (w6-w8 w10)*logtime / rows=2 columns=2 loess grid markerattrs=(size=5);
run;
title;
```

The PLOT statement specifies weighted residuals that correspond to the covariates **Race**, **Treatment**, **Sites**, and **Race*Site**. The ROWS= and COLUMNS= options specify the layout for paneling the plots. The LOESS option smooths the residuals, and the GRID option adds a grid. The residual display is shown in Figure 17.

Figure 17 Residual Display Produced by the SGSCATTER Procedure



Under the time-varying coefficients model, a plot of the weighted Schoenfeld residuals and their loess smooth should show no trend over time if the corresponding covariate has a proportional hazard. Three of the plots in Figure 17 support this assumption. The plot for **Treatment** indicates that the effect of the longer treatment is more pronounced in the earlier and later periods of follow-up. In fact, Hosmer, Lemeshow, and May (2008) concluded that this departure is not significant because the Wald test of **Treat** by **LogTime** interaction is not significant.

Create Hazard Function Plots by Using PROC PHREG

You might want to use the PHREG procedure to generate predicted survival curves for different sets of covariate values. You can specify the covariate values in the COVARIATES= data set in the BASELINE statement, and you can use the PLOTS= option in the PROC PHREG statement to display a survival curve for each row of covariates in this data set.

Suppose you want to compare the baseline survival curves for a 25-year-old participant who gets the short treatment and a 50-year-old participant who gets the long treatment. The following statements create a data set named **Row** in which **Age**=25 and **Treat**=0 for the first row and **Age**=50 and **Treat**=1 for the second row:

```
data Row;
   length AgeTreat $ 27;
   label agetreat = "Age-Treatment Combination";
   input Age BeckTotal nDrugFP1 nDrugFP2 IVHx Race Treat Site AgeTreat $ 27-55;
   datalines;
25 17 2.5 -2.2907 1 1 0 0 Age 25 and Short Treatment
50 17 2.5 -2.2907 1 1 1 0 Age 50 and Long Treatment
;
```

The other covariates have identical values for the two rows. The variable **AgeTreat** identifies the covariate sets.

The following statements produce a plot of the survival curves, which is shown in Figure 18:

```
proc phreg data=uis plots(overlay=stratum)=survival;
   ods select SurvivalPlot;
   class ivhx(ref="1");
   model time * censor(0) = age becktotal ndrugfp1 ndrugfp2 ivhx race treat
                            site age*site race*site;
   output out=out1 wtressch=w1-w10;
   baseline out=pred covariates=row survival=_all_ / rowid=AgeTreat;
run;
```

The PLOTS= option requests the survival plot, and the OVERLAY= option overlays the two curves in the same plot. The ROWID= option specifies that the values of **AgeTreat** be used to identify the curves. The OUT=PRED and SURVIVAL=_ALL_ options create an output data set that contains the survival function estimates.

Figure 18 Survival Plot Produced by the PHREG Procedure



The plot reveals that a 25-year-old participant in the short program is more likely to return to drug use than a 50-year-old in the long treatment program.

EXAMPLE 3: STATISTICAL GRAPHICS ANNOTATION

You can use statistical graphics (SG) annotation to add more components to graphs, including text, shapes, lines, images, and so on. You create a SAS data set that contains annotation instructions and then specify that data set in the SGANNO= option in PROC SGPLOT. You can create SG annotation data sets by using assignment and RETAIN statements and by processing other data sets. The following steps illustrate:

```
data anno(drop=name);
   DrawSpace = 'DataValue';
   if _n_ = 1 then do;
      Function = 'Text';      x1 = 2125;          y1 = 6;
      Rotate = -33;           Transparency = 0.8;
      TextColor = 'Red  ';    TextSize = 32;
      Label = 'Confidential Salary Information'; Width = 145;      output;
   end;

   set sashelp.baseball(keep=name logsalary crhits
                        rename=(logsalary=y1 crhits=x1)
```

```
                        where=(index(name, 'Rose')));
   function = 'Oval';   transparency = 0.5;
   Height = 5;          Width = 4;              rotate = 0;
   label = ' ';         textsize = .;           textcolor = ' ';  output;

   function = 'Text';   textcolor = 'Green';
   transparency = 0;    label = 'Pete Rose';    y1 + 0.4;
   width = 14;          textsize = 12;          height = .;       output;
run;

proc sgplot data=sashelp.baseball sganno=anno;
   title '1986 Baseball Salaries';
   scatter y=logsalary x=crhits;
   xaxis offsetmin=0.05 offsetmax=0.05;
   yaxis offsetmin=0.05 offsetmax=0.05;
run;

proc print noobs;
run;
```

The graph in Figure 19 contains a watermark across the diagonal and a label and oval that highlight an outlying point. The SG annotation data set is displayed in Figure 20.

Figure 19 Scatter Plot with SG Annotation



Figure 20 SG Annotation Data Set

DrawSpace	Function	x1	y1	Rotate	Transparency	TextColor	TextSize	Label	Width	Height
DataValue	Text	2125	6.00000	-33	0.8	Red	32	Confidential Salary Information	145	.
DataValue	Oval	4256	6.62007	0	0.5		.		4	5
DataValue	Text	4256	6.92007	0	0.0	Green	12	Pete Rose	14	.

All coordinates use data values as coordinates (`DrawSpace='DataValue'`). Several other coordinate options are available. The data set has three observations: `Function='Text'` (twice) and `Function='Oval'`. X and Y coordinates are specified in the **x1** and **y1** variables, respectively. Text is specified in the **Label** variable. The **Rotate** variable controls the angle of rotation for both the text and the oval. The roles of the other variables are obvious from their names.

SUMMARY

Table 2 summarizes ways in which you can use ODS Graphics in SAS 9.4.

Table 2 Summary of ODS Graphics Functionality

Graphical task	Audience	What do you use?	What should you learn?	
Create graphs in the context of statistical analyses	Statistical SAS users	Statistical procedures in SAS/STAT, SAS/ETS, SAS/QC, and Base SAS that support ODS Graphics	Specify ODS GRAPHICS ON; graphs are then created by default or by procedure options, which are documented in the procedure chapters.	Minimal graph syntax
Enhance specific graphs for a paper or presentation	Statistical and general SAS users	ODS Graphics Editor	How to request editable graphs, invoke the editor, use point-and-click features; see *SAS 9.4 ODS Graphics Editor: User's Guide*.	
Create stand-alone graphs for data exploration or for customized displays	Statistical and general SAS users	SGPLOT, SGPANEL, SGSCATTER procedures in Base SAS	SG procedure syntax; see *SAS ODS Graphics: Procedures Guide*.	
Change the overall appearance of graphs and tables	Statistical and general SAS users	ODS styles	STYLE= option in ODS destination statement	Some syntax
Save and manage graphs for papers and presentations	Statistical and general SAS users	ODS GRAPHICS options, ODS destination options	How to specify size and resolution; how to name and access image files	
Make persistent changes in graphs produced by statistical procedures (apply whenever you run your program)	Advanced SAS programmers	User modifications of graph templates that SAS provides	Basic features of the Graph Template Language and PROC TEMPLATE; see *SAS Graph Template Language: User's Guide*.	
Create a highly customized stand-alone graph	Advanced SAS programmers	ODS Graphics Designer	GUI for creating graph templates	Graphics programming
Create a highly customized stand-alone graph	Advanced SAS programmers	User-written graph templates	Graph Template Language, PROC TEMPLATE, and PROC SGRENDER; see *SAS Graph Template Language: User's Guide* and *SAS Graph Template Language: Reference*.	

REFERENCES

Hosmer, D. W., Jr., Lemeshow, S., and May, S. (2008). *Applied Survival Analysis: Regression Modeling of Time-to-Event Data*. 2nd ed. Hoboken, NJ: John Wiley & Sons.

Hsu, J. C. (1996). *Multiple Comparisons: Theory and Methods*. London: Chapman & Hall.

Hsu, J. C., and Peruggia, M. (1994). "Graphical Representation of Tukey's Multiple Comparison Method." *Journal of Computational and Graphical Statistics* 3:143–161.

Matange, S. (2013). *Getting Started with the Graph Template Language in SAS: Examples, Tips, and Techniques for Creating Custom Graphs*. Cary, NC: SAS Institute Inc.

Matange, S. (2016). *Clinical Graphs Using SAS*. Cary, NC: SAS Institute Inc.

Matange, S., and Heath, D. (2011). *Statistical Graphics Procedures by Example: Effective Graphs Using SAS*. Cary, NC: SAS Institute Inc.

ACKNOWLEDGMENTS

The authors are grateful to Anne Baxter, Jeff Cartier, Ed Huddleston, Ann Kuo, and Ying So for valuable assistance in the preparation of this paper.

RECOMMENDED READING

To get started using ODS Graphics and one of the statistical procedures listed on page 12, begin by using the examples and the "Syntax" section of the corresponding procedure chapter in the SAS/STAT, SAS/ETS, or SAS/QC user's guide or the Base SAS statistical procedures guide. Each chapter provides a section titled "ODS Graphics" that lists the available graphs. To get started using the statistical graphics procedures, see *SAS ODS Graphics: Procedures Guide*. Chapter 21, "Statistical Graphics Using ODS," in the *SAS/STAT User's Guide* covers the general functionality of ODS Graphics. Also see Chapter 22, "ODS Graphics Template Modification," in the same book.

Two new free online books discuss basic and advanced ODS Graphics examples. (Links are below.) *Basic ODS Graphics Examples* provides a gentle and parallel introduction to the GTL and the SG procedures. *Advanced ODS Graphics Examples* discusses SG annotation, attribute maps, advanced graph customization methods, axis tables, and other advanced topics.

ODS Graphics	http://support.sas.com/documentation/prod-p/grstat/
Basic Examples	http://support.sas.com/documentation/prod-p/grstat/9.4/en/PDF/odsbasicg.pdf
Advanced Examples	http://support.sas.com/documentation/prod-p/grstat/9.4/en/PDF/odsadvg.pdf
ODS	http://support.sas.com/documentation/cdl/en/statug/68162/HTML/default/viewer.htm#statug_ods_toc.htm
ODS Graphics	http://support.sas.com/documentation/cdl/en/statug/68162/HTML/default/viewer.htm#statug_odsgraph_toc.htm
Template Modification	http://support.sas.com/documentation/cdl/en/statug/68162/HTML/default/viewer.htm#statug_templt_toc.htm
Kaplan-Meier Plot	http://support.sas.com/documentation/cdl/en/statug/68162/HTML/default/viewer.htm#statug_kaplan_toc.htm

Also see Matange and Heath (2011) and Matange (2013, 2016).

CONTACT INFORMATION

Robert N. Rodriguez
SAS Institute Inc.
SAS Campus Drive
Cary, NC 27513
(919) 531-7650
Bob.Rodriguez@sas.com

Warren F. Kuhfeld
SAS Institute Inc.
SAS Campus Drive
Cary, NC 27513
(919) 531-7922
Warren.Kuhfeld@sas.com